Caustics Mapping: Explanation and Implementation in DirectX 9

	Figure 1: Images rendered using caustics mapping (from left to right): (a) under water caustics, (b) colored caustics, and (c) reflective caustics.


1. Introduction:
Real-time computer graphics has been greatly influenced by image-space algorithms, i.e., algorithms that operate on images of objects rather than the object geometry. For example, shadow mapping is a widely used image-space technique for rendering shadows by performing a simple depth test on rasterized pixels rather than doing expensive shadow-ray intersection tests with the scene geometry. Another popular example of image-space algorithms is displacement mapping. Essentially, image-space algorithms allow for a fast, approximative approach for rendering various optical effects that are prohibitively slow to produce using conventional geometric techniques, especially for real-time applications such as games. The reason why image-space techniques are faster is because they decouple the geometric complexity of a 3D scene from the render time. Therefore, in the case of shadow mapping, it does not matter if there are ten or one hundred objects in the scene, the algorithm will take more or less the same amount of time to render shadows. Whereas if the shadows were being rendered using ray-casting, the time costs will increase significantly as more objects are introduced into the scene. The second advantage of image-space algorithms is that they are very well suited for the current programmable graphics hardware since GPUs are great at brute force per-pixel operations. 
As hardware becomes more and more powerful, games start to pack in more and more eye-popping graphics effects, thanks to a very active graphics research community. Recent research developments have introduced a new image-space technique for rendering caustics in real-time entitled “caustics mapping”. This article is a tutorial style explanation of the technique and serves as a stepwise guide to implementing caustics mapping.

	Figure 2: Caustics formation: light rays from the source get focused on a single point on the receiver surface due to the refractive object.


2. What are caustics?
Before we dive into the details of rendering caustics, let’s first define what they actually mean. Simply put, caustics are complex patterns of shimmering light that can be seen on surfaces in presence of reflective or refractive objects, for example those formed on the floor of a swimming pool in sunlight. A more formal definition is that whenever multiple rays of light converge on the same point on a surface, they cause that region to become relatively brighter than its surrounding regions. These non-uniform distributions of bright and dark regions are known as caustics. The convergence of the light rays is generally caused by reflective or refractive objects, which bend the rays causing them to change their direction and focus on a single point (see Figure 2). The caustics mapping algorithm simulates this light transport using an approximative image-space approach to render visually convincing caustics in real-time.

	Figure 3: The caustics mapping algorithm: (top) the creation of the caustics map texture using photon splatting, and (bottom) rendering caustics on receiver geometry by looking up the caustics map.


3. The Caustics Mapping Algorithm

The caustics mapping algorithm runs entirely on the GPU, therefore freeing up the CPU to perform the numerous other tasks required in games such as the AI. Conversely, that also implies the reader should have some knowledge of vertex and pixel shader programming in order to fully understand this algorithm. The graphics API and shader language used in this implementation are Microsoft DirectX 9.0 and HLSL; however it is equally convenient to implement the algorithm using OpenGL and GLSL or CG. Lastly, a graphics card with support for shader model 3.0 is required since the algorithm performs texture lookups in the vertex shader.
We are now ready to look at the caustics mapping algorithm. Without loss of generality, caustics from a refractive object will be considered throughout this tutorial. The use of reflective objects does not require any modifications to the overall algorithm, and therefore can be substituted in place of the refractive objects if desired. 
The algorithm is conceptually quite simple and intuitive. It starts with tracing light through the refractive object. The footprints of the refracted light are then collected onto an image plane facing the light source to create a caustics map. Caustics are then rendered onto surfaces by looking up the caustics map. Figure 3 shows a schematic of the algorithm. 

Straight away we run into a few major problems that need to be addressed. First, we need to compute the intersection points of the refracted light rays with the scene geometry to determine where the caustics will form. This has to be done fast enough so that the real-time performance is not compromised. For now let’s assume that we are given some magic function that takes a refracted light ray and returns the intersection point at virtually no cost. We can therefore shoot a light ray from the light source to each point on the surface of the refractive object, perform the refraction, and then use the intersection function to compute where the refracted light ray intersects in the rest of the 3D world. For each ray, a certain amount of light is deposited in the caustics map at the location of projection of intersection point. So when several refracted light rays intersect at the same point, multiple light deposits will be accumulated in that region causing it to get brighter and thus forming caustics. However, this accumulation process presents another challenge. Since we will be implementing the algorithm on the GPU, we cannot randomly write to pixels; they can only be written in the order that they are rasterized. Therefore, this means that if we want to write to certain pixels more than once, they need to be rasterized more than once. The caustics mapping algorithm solves this problem using “vertex splatting” (also known as point splatting), which simply displaces the vertices of the refractive object along the refracted light direction and renders them as point primitives. As a result of the displacement, multiple vertices can end up in the same position thus providing the repeated rasterization of the same pixels that is required for the light accumulation. Additive blending is used to sum up the contribution from each individual fragment. 

We are still left with one small problem. Since the algorithm works using vertex splatting, the quality of the caustics produced depends heavily on the number of vertices of the refractive object. For example, if we want to render caustics from a glass cube, which would generally be modeled using only eight vertices, we would end up with random, disjoint patches of light. A simple workaround is to tessellate the mesh so that it contains a large number of vertices. However, using highly tessellated meshes in games and other real-time applications is not desired, especially for simple objects such as a cube. A more elegant solution is to create an auxiliary mesh, known as the refractive vertex grid, containing a set of vertices evenly distributed over the surface of the refractive object visible from the light source. This mesh can then be used for the splatting process instead of the original refractive object mesh. 
3.1. Implementation

 Let us now look at caustics mapping from an implementation oriented perspective. Following are the main steps of the algorithm:

Step 1: Setup refractive vertex grid.
Step 2: Create caustics map.

Step 3: Render caustics on receiver geometry using the caustics map.

Users familiar with shadow mapping might notice a striking structural similarity, with the exception of the first step. Let’s consider each of the steps in some detail. First, a “refractive vertex grid” has to be created. As mentioned earlier, this is simply a set of vertices that will be “splatted” wherever rays of light hit the receiver geometry. Splatting means exactly what it sounds like. Think of it as a paint ball being thrown at a wall, and thus creating a splat on it. In our case, the paint ball is a vertex which represents a photon (or a number of photons), and the wall is the receiver geometry, i.e., surfaces on which the caustics can be formed. The idea is that if multiple vertices, traveling along the light rays, end up getting splatted at the same point, that region will become brighter. Recall that this is exactly what we wanted to simulate for rendering caustics. The results of the splatting are stored in a texture called the caustics map. Then in the final rendering stage, each point on the receiver geometry is projected into the caustics map texture to determine the amount of caustics it receives, if any.
It is time to get down and dirty with the implementation details. We will reiterate through the steps of the caustics mapping algorithm and consider what needs to be done at each step. Before we begin, let’s define the scene that we will be rendering. First, we need a refractive (or reflective) object that will cause the light rays to converge, thus forming caustics. Second, we need some receiver geometry on which the caustics will be formed. The choice of these objects is left solely to the user; the algorithm places no limitation on the type of 3D geometry used. A good starter selection would be to use a sphere for the refractive object and a flat plane underneath it as the caustic receiver geometry. Lighting is provided by a point light source (the algorithm currently does not support environment or area lighting). Once that is decided, we can start with the implementation of the first step of the algorithm: setting up the refractive vertex grid. 
	[image: image7.png]Surface point locations

I
I /

- I /
Splat Iocatlons’: 4
/





	Figure 4: Final splat locations for the vertices in the refractive vertex grid.


The goal of the refractive vertex grid is to give us a uniform distribution of points on the surface of the refractive object visible from the light source. This can be achieved by rendering the refractive object from the light’s point of view onto a texture of a certain resolution. Instead of outputting color, the 3D world positions are outputted at each pixel. Then, if we create a rectangular vertex grid of the same resolution, we end up with a one to one correspondence between the pixels of the texture and the vertices of the grid. And since the texture contains 3D positions of the points on the surface of the refractive object instead of color, we essentially end up with a 3D position for each grid vertex. Mission accomplished; we have a set of vertices evenly distributed over the surface of the refractive object and can now proceed with the rest of the algorithm. However, we can take this a step further. Recall that the ultimate objective of the refractive vertex grid is to splat the vertices at the points of intersection of the refracted light rays and the receiver geometry. So instead of points on the surface of the refractive object, why not render the final intersection points onto the texture and apply those to the grid (See Figure 4)? And that is exactly what the algorithm does. 
The vertex grid is created and initialized as follows:
	#define PP_RES 100

// Refractive vertex grid

IDirect3DVertexBuffer9* pRefractVertBuffer;

struct myVertex

{


D3DXVECTOR3 pos;


D3DXVECTOR2 tc;

};

#define D3DFVF_MYVERT (D3DFVF_XYZ | D3DFVF_TEX2 | D3DFVF_TEXCOORDSIZE2(1))

	


	// Create vertex buffer for the refractive vertex grid

myVertex * vertices = new myVertex[PP_RES * PP_RES];

float delta = 2.0/float(PP_RES-1.0);

float deltat = 1.0/float(PP_RES-1.0);

for(int i=0;i<PP_RES;i++)

{


for(int j=0;j<PP_RES;j++)


{



vertices[i*PP_RES + j].pos = D3DXVECTOR3(-1.0+delta*j,1.0-

                                                     delta*i,1.0);



vertices[i*PP_RES + j].tc = D3DXVECTOR2(deltat*j,deltat*i);



}


}

}
pd3dDevice->CreateVertexBuffer(sizeof(myVertex)*PP_RES*PP_RES, 0,       

                               D3DFVF_MYVERT, D3DPOOL_MANAGED,

                               &pRefractVertBuffer, NULL);


	// fill the vertex buffer with the refractive vertex grid

void *pData;

pRefractVertBuffer->Lock(0,0,&pData,0);

memcpy(pData,vertices,sizeof(myVertex)*PP_RES*PP_RES);

pRefractVertBuffer->Unlock();

delete [] vertices;


The code above creates a DirectX vertex buffer for the refractive vertex grid. The grid is planar, with a constant z coordinate equal to 1.0, and the x and y coordinates ranging from -1.0 to 1.0. Each vertex in the grid also has associated texture coordinates u and v, which range from 0 to 1.0. PP_RES defines the resolution of the refractive vertex grid. The sharpness and quality of the rendered caustics depends on this resolution. The value can be adjusted by the user to achieve the desired balance between speed and image quality.
Now that the refractive vertex grid has been setup, we need to determine the positions where the vertices will be splatted to create the caustics map. These positions are the intersection points of the refracted lights rays with the receiver geometry. However, since we cannot afford to perform ray-geometry intersection tests, we will use an image-space approximation technique. Remember that magic intersection function that we assumed to possess? Well, its time to define it now. Following are the steps that need to be taken:
Step 1: From the light’s point of view, render the receiver geometry (do not render the refractive object) onto a texture. Instead of outputting color at every pixel, the 3D coordinates in world space are output. We will refer to this as the “positions texture”.
Step 2: From the light’s point of view, render the refractive object onto a texture. At every pixel, compute the refracted light ray and estimate intersection point with the receiver geometry using the positions texture. Output the 3D coordinates in world space of the intersection point.

Step 1 is pretty straightforward; it will consist of simple shader programs that you have probably written a thousand times. The only difference is instead of outputting color at each pixel, you will output the interpolated 3D world positions. Step 2 is a little more involved. Following are the vertex and pixel shaders corresponding to Step 2:

Vertex Shader:

	float4x4 g_mWorld;
// World transformation matrix

float4x4 g_mLightView;
// View matrix for light

float4x4 g_mLightProj;
// Projection matrix for light

// vertex input and output structures

struct VS_IN

{

    float4 pos : POSITION;    // 3D position
    float3 normal : NORMAL0;  // vertex normal
};

struct VS_OUT

{

    float4 pos    : POSITION;  // position in clip space

    float4 posi   : TEXCOORD0; // interpolated position in world space

    float3 normal : TEXCOORD1; // interpolated normal in world space

};

// The vertex shader
VS_OUT vsLightRender(VS_IN input)

{


float4x4 mLightViewProjection = mul(g_mLightView, g_mLightProj);


VS_OUT output;


output.posi = mul(input.pos,g_mWorld);


output.normal = mul(input.normal,g_mWorld);


output.pos = mul(output.posi,mLightViewProjection);

return output;

}

	


Pixel Shader:

	// Compute texture coordinates by projecting the given 3D position in // world space

float2 getTC(float4x4 mVP, float3 wpos)

{


float4 texPt = mul(float4(wpos,1),mVP);


float2 tc = float2(0.5*(texPt.xy/texPt.w) + float2( 0.5, 0.5 ));

    tc.y = 1.0f - tc.y;

    return tc;

}

// Estimate ray/geometry intersection point

float rayGeo(float3 pos, float3 dir, float4x4 mVP, sampler posTex, int num_iter)

{


float2 tc = float2(0.0,0.0);


// initial guess


float dist = 1.0;


for(int i=0;i<num_iter;i++)


{



float3 pos_p = pos + dir*dist;



tc = getTC(mVP, pos_p);



float4 newPos = tex2D(posTex, tc);



dist = distance(newPos.xyz/newPos.w, pos);


}


return dist;

}
// The pixel shader

float4 psIntersectPts(VS_OUT input) : COLOR

{


float3 posi = input.posi.xyz/input.posi.w;


float3 normal = normalize(input.normal);


float4x4 mVP = mul(g_mLightView, g_mLightProj);


// Compute refracted ray direction

float3 lightDir = normalize(posi - g_lightPos);


float3 refrDir = normalize(refract(lightDir,normal,g_refIndex));


// Get distance to ray/geometry intersection estimation point


float dist = rayGeo(posi, refrDir, mVP, posTexSampler, 1);


// Compute the intersection point

float3 intPt = posi + refrDir*dist;
      return float4(intPt,1);
}

	


The pixel shader utilizes the magic intersection estimation function, rayGeo. This function implements an iterative root finding method similar to the Newton-Raphson method. The basic idea is that the problem of finding the intersection point is posed as the problem of finding the root of a mathematical function. The function in our case is defined by the positions texture. Therefore, the root of our function is where a given light ray intersects the positions texture, which is essentially the intersection point that we are trying to find. There are a number of root-finding methods out there, such as the Secant method, Bisection method, etc. and they differ in accuracy, robustness, and the number of iterations it takes to converge to the solution. Any one of these methods can be used for the intersection estimation. The authors of caustics mapping decided to go with a variant of the Newton-Raphson method which converges faster than the others. For further details on the convergence and accuracy of this function, please consult the caustics mapping paper [SKP2006]. 
The rayGeo function returns an estimate of the distance to the intersection point of the ray along the given direction, dir, originating from the point, pos. It uses the positions texture (posTexSampler) created in Step 1. The parameter num_iter specifies the number of iterations to be performed. Generally, each successive iteration increases the accuracy of the approximation. However, for the purpose of rendering caustics just a single iteration is sufficient. 
After this step, we end up with a texture containing the intersection points of the refracted light rays and the receiver geometry. The hard part is over, everything after this is just simple point rendering with texture mapping. In order to create the caustics map, all we need to do now is splat the vertices from the refractive vertex grid at these intersection points. We do this by rendering the grid as point primitives using additive alpha blending onto a texture, i.e. the caustics map. In DirectX, point primitives are rendered as screen-aligned quads. Therefore, if we use them for splatting, the splats will look like squares and create some unwanted visual artifacts. However, we can apply an alpha mask with a Gaussian falloff to the point primitives so that they have a nice circular shape with a smooth gradient at the edges. In DirectX, the PointSpriteEnable state must be set in order to use point primitive texturing. Figure 5 shows an example texture that can be used as the alpha mask. The white color indicates total transparency whereas the black color indicates total opaqueness. 

	Figure 5: Alpha mask for point splat.


The vertex and pixel shaders for this step are as follows:
Vertex Shader:
	// Vertex input and output structures

struct VS_IN1

{


float4 pos : POSITION;


float2 tc : TEXCOORD0;

};
struct VS_OUT1

{


float4 pos : POSITION;


float3 light_int : COLOR0;


float2 tc : TEXCOORD0;

};
// The vertex shader

VS_OUT1 vsCaustics(VS_IN1 input)

{



// look up the intersection position


float4 intPt = tex2Dlod(ppPosSampler, float4(input.tc,0,0));


// set the output value

             float4x4 mLightViewProj = mul(g_mLightView, g_mLightProj);


output.pos = mul(float4(intPt,1),mLightViewProj);


// compute the light intensity


output.light_int = float3(1000,1000,1000)*(4.0/g_ppRes);


return output;

}

	


Pixel Shader:

	// Caustics shader

float4 psCaustics(VS_OUT1 input) : COLOR

{


// look up alpha mask


float4 tcol = tex2D(pixelSampler,input.tc);


return float4(input.light_int,1-tcol.r);

}




In the vertex shader, the intersection point is looked up from the texture we created in the previous pass using the image-space intersection approximation. This point is where the point primitive will be rendered. However, since the point stored in the texture is in world space, we must first project it into the light’s view space so that it can be stored in the caustics map. The vertex shader output also consists of the light contribution from the current vertex. This is computed by dividing the total light intensity by the resolution of the refractive vertex grid. In the pixel shader, the interpolated light contribution is output which gets accumulated when multiple point primitives overlap the same pixels. Remember, you must have additive alpha blending enabled for this accumulation to take place!

	Figure 6: An example caustics map (left) and the actual 3D scene rendered using the map (right).


At the end of this render pass, we will have our caustics map texture. Figure 6 shows an example of a caustics map texture. Now all that’s left to do is performing the final rendering of our 3D scene (with caustics of course!) and display it on the screen. For this step, you will render the scene as you normally would. The only exception is that when computing lighting on the receiver surfaces, other than the point light you have an additional light source: the caustics map. Therefore, the total incident light at a point is given by the light source and the caustics falling on that point, if any. So for example, a diffuse surface with caustics will be shaded as follows:
	// Caustics getter

float3 get_caustics(float3 pos)

{


float4x4 mLightViewProj = mul(g_mLightView, g_mLightProj);


float4 texPt = mul(float4(pos,1),mLightViewProj);


float2 tc = float2(0.5*(texPt.xy/texPt.w) + float2( 0.5, 0.5 ));

      tc.y = 1.0f - tc.y;


float4 caustic = tex2D(causticsSampler,tc);


float dec = saturate(dot(normalize(-g_lightPos),normalize(pos)));


if(dec == 0) caustic = caustic*0;


return pow(float3(caustic.rgb),1.3);

}

// Diffuse shader with caustics
float4 psDiffuse(VS_OUT input) : COLOR

{

    // compute Lambertian diffuse term

    float3 diffuse = float3(1,1,1)*

                       saturate(dot(normalize(input.normal),

                       normalize(g_lightPos -            

                                 (input.posi.xyz/input.posi.w))));

    // compute final color
    float3 col = (diffuse + 

                  diffuse*get_caustics(input.posi))*g_diff.xyz;

    return float4(col,1);

}


The get_caustics function takes the current 3D point in world-space, projects it into the light’s view space, and looks up the caustics map texture. It returns the amount of light falling on that point due to caustics. In the pixel shader, diffuse shading is performed by using the light from the point light source as well as the caustics. The g_diff variable is the diffuse material color of the objecting being rendered. 
You should now be looking at a pretty picture with nice caustics, assuming everything went well in all the steps that we covered. If things don’t look quite the way you want them to look, try adjusting some parameters here and there (such as the refractive index of the refractive object, or the light intensity in the caustics map creation step). In my demo application, I generally code in some sliders that I can use to tweak the various parameters until the desired results are achieved. Also, keep in mind that the code presented above was written to make things easy to explain and therefore is not the most optimal implementation. It is left to the user to come up with hacks and trick to achieve killer frame-rates. 
4. Conclusion

In this article, we went through the steps of implementing a new technique for rendering caustics in real-time called caustics mapping. The algorithm utilizes a caustics map, which is created using vertex splatting at the intersection points of the refracted/reflected light rays and the receiver geometry. The intersection points are computed using a fast, approximative technique based on the Newton-Raphson root-finding method and it operates in image-space. This enables the algorithm to achieve high frame rates which are suitable for games and other interactive graphics applications. Furthermore, the algorithm runs entirely on the GPU so the CPU is free to perform other operations such as AI in games. Lastly, caustics mapping can easily be integrated into graphics and game engines, very much like shadow mapping, since it does not hinder or affect any other rendering process. 

